SAS® GLOBAL FORUM 2016

IMAGINE. CREATE. INNOVATE.

Portfolio Optimization with Discontinuous Constraint

#SASGF

Portfolio Optimization with Discontinuous Constraint

Robert Spatz, Taras Zlupko

University of Chicago

ABSTRACT

Optimization models require continuous constraints to converge. However, some real life problems are better described by models which incorporate discontinuous constraints. A common type of such discontinuous constraints becomes apparent when regulation-mandated diversification requirement is implemented in investment portfolio model. Generally stated the requirement postulates that the aggregate of investments with individual weights exceeding certain threshold in the portfolio should not exceed some predefined total within the portfolio. This format of the diversification requirement may be defined by the rules of any specific portfolio construction methodology and is commonly imposed by the regulators. We discuss the impact of this type of discontinuous portfolio diversification constraint on the portfolio optimization model solution process and develop a convergent approach. The latter includes a sequence of definite series of convergent non-linear optimization problems and is presented in the framework of the PROC OPTMODEL modeling environment. The approach discussed has been used in constructing investable equity indexes.

MODEL

$$min \sum_{i=1}^{N}(w_i - x_i)^2 \tag{1}$$

subject to

$$\sum_{i=1}^{N} x_i = 1 \tag{2}$$

$$0 \le x_i \le 0.25 \tag{3}$$

$$\sum_{i:\, x_i > 0.05} x_i \le 0.5 \tag{4}$$

where

w_i – weight of i-th security in original portfolio

x_i – weight of i-th security in regulation-compliant portfolio

N – number of securities in portfolio

SOLUTION

```
proc optmodel;
   number iw{1..28};
   read data input into [_n_] iw;
   /*model description*/
   var x{i in 1..28} >=0.0 <=0.25;
   min z = sum{i in 1..28}((iw[i]-x[i])**2);
   con c1:sum{i in 1..28} (if x[i]>=0.05 then x[i] else 0) <=0.5;
   con c2:sum{i in 1..28} x[i]=1.0;
   solve;
quit;
```

Portfolio Optimization with Discontinuous Constraint

Robert Spatz, Taras Zlupko

University of Chicago

SOLUTION: MODEL (1)-(4)

	Label1	cValue1	nValue1
1	Objective Sense	Minimization	.
2	Objective Function	z	.
3	Objective Type	Quadratic	.
4			.
5	Number of Variables	28	28.000000
6	Bounded Above	0	0
7	Bounded Below	0	0
8	Bounded Below and Above	28	28.000000
9	Free	0	0
10	Fixed	0	0
11			.
12	Number of Constraints	2	2.000000
13	Linear LE (<=)	0	0
14	Linear EQ (=)	1	1.000000
15	Linear GE (>=)	0	0
16	Linear Range	0	0
17	Nonlinear LE (<=)	1	1.000000
18	Nonlinear EQ (=)	0	0
19	Nonlinear GE (>=)	0	0
20	Nonlinear Range	0	0

Problem Summary

	Label1	cValue1	nValue1
1	Solver	NLP/INTERIORPOINT	.
2	Objective Function	z	.
3	Solution Status	Iteration Limit Reached	.
4	Objective Value	0.0571642358	0.057164
5	Iterations	5000	5000.000000
6			.
7	Optimality Error	0.004459369	0.004459
8	Infeasibility	9.8461297E-7	0.000000985

Solution Summary

```
        4997         0.05716516   0.0000000006087          4.95404164
        4998         0.05716516   0.0000000006087          4.95404164
        4999         0.05716516   0.0000000006087          4.95404163
        5000         0.05701099        0.05815258          0.05815258
NOTE: Maximum number of iterations reached.
NOTE: Objective = 0.057010991.
NOTE: Objective of the best feasible solution found = 0.0571642358.
NOTE: The best feasible solution found is returned.
```

Solver Log at Exit

Portfolio Optimization with Discontinuous Constraint

Robert Spatz, Taras Zlupko

University of Chicago

OBJECTIVE AND INFEASIBILITY: MODEL (1)-(4)





Portfolio Optimization with Discontinuous Constraint

Robert Spatz, Taras Zlupko

University of Chicago

DISCONTINUOUS CONSTRAINT

$$\sum_{i:\,x_i>0.05} x_i \le 0.5 \qquad (4)$$

INSIGHT

- While i in the constraint (4) can include different securities in a portfolio, no more than nine securities in any given portfolio can have weights greater than 0.05 ("large" securities) and have the sum of their weights be less than or equal to 0.5 for the constraint (4) to be true.
- The rank order of weights does not change in a least squares minimization (i.e. if $w_i < w_j$, then $x_i < x_j$). Therefore the candidates for the largest securities in the regulation-compliant portfolio can be selected from a sequence of initial portfolio weights sorted in descending order.

RE-STATED MODEL

$$min \left[\sum_{l=1}^{L}(w_l - x_l)^2 + \sum_{s=1}^{N-L}(w_s - x_s)^2 \right] \qquad (1a)$$

subject to

$$\sum_{l=1}^{L} x_l + \sum_{s=1}^{N-L} x_s = 1 \qquad (2a)$$

$$0 \le x_s \le 0.05 \qquad (3a)$$

$$0 \le x_l \le 0.25 \qquad (3b)$$

$$\sum_{l=1}^{L} x_l \le 0.5 \qquad (4a)$$

where:

x_l – weight of l-th security that may be "large" in regulation-compliant portfolio

w_l – weight of l-th security in the original portfolio

x_s – weight of s-th security that must be "small" in regulation-compliant portfolio

w_s – weight of s-th security in the original portfolio

w_i – weights of securities in the original portfolio are sorted in descending order

N – number of securities in portfolio

L – number of securities that may be "large" in in regulation-compliant portfolio. L can range from zero to nine. When L=0, the terms based on L treated as zero

Model (1a)-(4a) is reflective of the state when both initial and optimized portfolios are split into "large" and "small".

Although the maximum number of "large" securities as defined above in the regulation-compliant portfolio is known and equals nine, the unknown is the actual number of securities that will need to be "large" in the optimized portfolio in order to achieve the minimum of the objective function. Therefore model (1a)-(4a) is solved for each L.

Portfolio Optimization with Discontinuous Constraint

Robert Spatz, Taras Zlupko

University of Chicago

SOLUTION: MODEL (1a)-(4a) WHEN L=2

1. Sort original portfolio weights in descending order.
2. Designate “large” and “small” securities based on L.
3. Define and solve the model using PROC OPTMODEL.
4. Combine optimized portfolio weights into single dataset.

```
/*1.Sort original portfolio weights in descending order*/
proc sort data=d.input; by descending iw;
run;

/*2. Designate “large” and “small” securities when L=2*/
data d.input_l (keep=idxl iwl) d.input_s (keep=idxs iws);
   set d.input;
      length idxl iwl idxs iws 8;
         if (_N_ <= 2) then do;
                idxl  =  idx;
                iwl   =  iw;
                 output d.input_l;
                                 end;
         else do;
                idxs  =  idx;
                iws   =  iw;
                output d.input_s;
        end;
run;
```

```
/*3. Define and solve the model*/
proc optmodel;
   ods output ProblemSummary=exps SolutionSummary=exss;

/*define initial values*/
number iws{1..26};
number iwl{1..2};
read data d.input_s into [_n_] iws;
read data d.input_l into [_n_] iwl;

/*declare variables, objective and constraints and solve*/
var xs{i in 1..26} >=0.0 <=0.05;
var xl{i in 1..2}  >=0.0 <=0.25;
min z=sum{i in 1..26}((iws[i]-xs[i])**2)+sum{i in 1..2}((iwl[i]-xl[i])**2);
con c1:sum{i in 1..2}  xl[i] <=0.5;
con c2:sum{i in 1..26} xs[i] + sum{i in 1..2} xl[i] = 1.0;
solve;

/*4. Combine regulation-compliant portfolio weights into single dataset*/
create data d.output_large from [_n_] iwl xl;
create data d.output_small from [_n_] iws xs;
quit;
```

Portfolio Optimization with Discontinuous Constraint

Robert Spatz, Taras Zlupko

University of Chicago

SOLUTION: MODEL (1a)-(4a)

	Label1	cValue1	nValue1
1	Objective Sense	Minimization	.
2	Objective Function	z	.
3	Objective Type	Quadratic	.
4			.
5	Number of Variables	28	28.000000
6	Bounded Above	0	0
7	Bounded Below	0	0
8	Bounded Below and Above	28	28.000000
9	Free	0	0
10	Fixed	0	0
11			.
12	Number of Constraints	2	2.000000
13	Linear LE (<=)	1	1.000000
14	Linear EQ (=)	1	1.000000
15	Linear GE (>=)	0	0
16	Linear Range	0	0

Problem Summary, L=2.

	Label1	cValue1	nValue1
1	Solver	NLP/INTERIORPOINT	.
2	Objective Function	z	.
3	Solution Status	Optimal	.
4	Objective Value	0.0571645707	0.057165
5	Iterations	6	6.000000
6			.
7	Optimality Error	5E-9	5E-9
8	Infeasibility	1.5079629E-9	1.5079629E-9

Solution Summary, L=2.

L	Objective Value	Iteration	Solver Time	Solution Status
0	0.2568828	6	0.016	Optimal
1	0.1154255	4	0.000	Optimal
2	0.0571646	6	0.031	Optimal
3	0.0603897	4	0.000	Optimal
4	0.0617770	6	0.000	Optimal
5	0.0630380	5	0.016	Optimal
6	0.0641847	5	0.015	Optimal
7	0.0654369	6	0.016	Optimal
8	0.0664391	6	0.015	Optimal
9	0.0674577	5	0.016	Optimal

Summary of Overall Results

Solution at each L is completed in six or less iterations. All runs arrived to optimal solution. Best objective value is at L=2.

Portfolio Optimization with Discontinuous Constraint

Robert Spatz, Taras Zlupko

University of Chicago

SOLUTION: MODEL (1a)-(4a)





Portfolio Optimization with Discontinuous Constraint

Robert Spatz, Taras Zlupko

University of Chicago

CONCLUSIONS

- A solution of the optimization model that includes discontinuous constraint has been discussed. This type of model arises when constructing investment portfolios that comply with certain regulatory requirements. The solution is based on the insight about the weight composition of the regulatory-compliant portfolio and properties of objective function employed by the model. This allows to re-state the initial model and replace discontinuous constraint with a continuous linear inequality constraint. The solution is found by solving the re-stated model while varying number of securities that may exceed weight threshold as defined by the regulation.

- Example given shows that the proposed solution efficiently converges to optimality and significantly improves solver performance. While solving the initial model, the solver exits on reaching maximum number of iterations. The re-stated model is solved by reaching optimal solution with much fewer iterations of the solver.

- Discussion of the model, discontinuous constraint, solution and examples has been conducted in the context of specific regulatory requirements. The solution approach is applicable in other contexts and can be generalized to be useful in other cases which are described by relevant type of model and constraints .

REFERENCES

Directive 2009/65/EC of the European Parliament and of the Council of 13 July 2009 on the coordination of laws, regulations and administrative provisions relating to undertakings for collective investment in transferable securities (UCITS), Chapter VII, Obligations concerning the investment policies of UCITS, Article 52. Accessed March 10, 2016. http://eur-lex.europa.eu/LexUriServ/LexUriServ.do?uri=OJ:L:2009:302:0032:0096:en:PDF

IRS Code Title 26 section 851(b)(3). Accessed March 10, 2016. http://www.gpo.gov/fdsys/pkg/USCODE-2010-title26/pdf/USCODE-2010-title26-subtitleA-chap1-subchapM.pdf

Lobo, M.G., Fazel, M. and S. Boyd. 2007. “Portfolio optimization with linear and fixed transaction costs.” *Annals of Operations Research. Financial Optimization,* 152,1:341–365.

SAS Institute Inc. 2014. SAS/OR® 13.2 User’s Guide: Mathematical Programming. Cary, NC: SAS Institute Inc.

Spatz, R. and T. Zlupko. 2015. “Portfolio Construction with OPTMODEL.” *Proceedings of the SAS Global Forum 2015 Conference*, Dallas, TX. Available at http://support.sas.com/resources/papers/proceedings15/3225-2015.pdf

SAS® GLOBAL FORUM 2016

IMAGINE. CREATE. INNOVATE.

LAS VEGAS | APRIL 18-21

#SASGF